

...una invitación a entrar en el maravilloso mundo de la programación...

[Portada](#)
[Noticias](#)
[Descargar](#)
[Documentación](#)
[Foros](#)

Mayores

Pseudocódigo: [Flexible...](#) [Estricto...](#) | Diagramas de flujo: [Clásico...](#) [Nassi-Schneiderman...](#) | [Más ejemplos...](#)

// Busca Los dos mayores de una Lista de N datos

Proceso Mayores

// primero se declara un arreglo de 200 elementos

Dimension datos[200]

// luego se pide al usuario ingresar La cantidad de datos,

// que no debera ser mayor a 200

Escribir "Ingrese la cantidad de datos (de 2 a 200):"

Leer n

// se Leen uno por uno Los 200 datos y se Los guarda en el arreglo

Para i<-1 **Hasta** n **Hacer**

Escribir "Ingrese el dato ",i,":"

Leer datos[i]

FinPara

// se comparan Los dos primeros y se toman como may1 (el mayor de Los

// dos) y may2 (el segundo mayor).

Si datos[1]>datos[2] **Entonces**

may1<-datos[1]

may2<-datos[2]

Sino

may1<-datos[2]

may2<-datos[1]

FinSi

// se recorren Los demas elementos buscan si hay alguno mayor que may1 o may2

Para i<-3 **Hasta** n **Hacer**

Si datos[i]>may1 **Entonces** *// si hay un valor mayor que may1*

may2<-may1 *// como may1 era el más grande, pasa a estar en segundo lugar*

may1<-datos[i] *// y el nuevo dato toma el primer puesto (mayor de todos)*

Sino *// si no era mas grande que may1, todavia puede ser mas grande que may2*

Si datos[i]>may2 **Entonces** *// si supera al segundo mayor que teniamos*

may2<-datos[i] *// se Lo guarda como segundo mayor*

FinSi

FinSi


FinPara

// se muestran Los resultados

Escribir "El mayor es: ",may1

Escribir "El segundo mayor es: ",may2

FinProceso



Powered by:  SOURCEFORGE